# NAG Library Routine Document

# F07TEF (DTRTRS)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

F07TEF (DTRTRS) solves a real triangular system of linear equations with multiple right-hand sides, $AX = B$ or $A^{\mathrm{T}}X = B$.

## 2 Specification

```
SUBROUTINE F07TEF (UPLO, TRANS, DIAG, N, NRHS, A, LDA, B, LDB, INFO)

INTEGER            N, NRHS, LDA, LDB, INFO
REAL (KIND=nag_wp) A(LDA,*), B(LDB,*)
CHARACTER(1)       UPLO, TRANS, DIAG
```

The routine may be called by its LAPACK name ***dtrtrs***.

## 3 Description

F07TEF (DTRTRS) solves a real triangular system of linear equations $AX = B$ or $A^{\mathrm{T}}X = B$.

## 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Higham N J (1989) The accuracy of solutions to triangular systems *SIAM J. Numer. Anal.* **26** 1252–1265

## 5 Parameters

1: UPLO – CHARACTER(1) *Input*

*On entry*: specifies whether $A$ is upper or lower triangular.

UPLO = 'U'
$A$ is upper triangular.

UPLO = 'L'
$A$ is lower triangular.

*Constraint*: UPLO = 'U' or 'L'.

2: TRANS – CHARACTER(1) *Input*

*On entry*: indicates the form of the equations.

TRANS = 'N'
The equations are of the form $AX = B$.

TRANS = 'T' or 'C'
The equations are of the form $A^{\mathrm{T}}X = B$.

*Constraint*: TRANS = 'N', 'T' or 'C'.

3: DIAG – CHARACTER(1) *Input*

*On entry*: indicates whether $A$ is a nonunit or unit triangular matrix.

DIAG = 'N'
$A$ is a nonunit triangular matrix.

DIAG = 'U'
$A$ is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.

*Constraint*: DIAG = 'N' or 'U'.

4: N – INTEGER *Input*

*On entry*: $n$, the order of the matrix $A$.

*Constraint*: $\mathrm{N} \geq 0$.

5: NRHS – INTEGER *Input*

*On entry*: $r$, the number of right-hand sides.

*Constraint*: $\mathrm{NRHS} \geq 0$.

6: A(LDA,$*$) – REAL (KIND=nag_wp) array *Input*

**Note**: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

*On entry*: the $n$ by $n$ triangular matrix $A$.

If UPLO = 'U', $A$ is upper triangular and the elements of the array below the diagonal are not referenced.

If UPLO = 'L', $A$ is lower triangular and the elements of the array above the diagonal are not referenced.

If DIAG = 'U', the diagonal elements of $A$ are assumed to be 1, and are not referenced.

7: LDA – INTEGER *Input*

*On entry*: the first dimension of the array A as declared in the (sub)program from which F07TEF (DTRTRS) is called.

*Constraint*: $\mathrm{LDA} \geq \max(1, \mathrm{N})$.

8: B(LDB,$*$) – REAL (KIND=nag_wp) array *Input/Output*

**Note**: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

*On entry*: the $n$ by $r$ right-hand side matrix $B$.

*On exit*: the $n$ by $r$ solution matrix $X$.

9: LDB – INTEGER *Input*

*On entry*: the first dimension of the array B as declared in the (sub)program from which F07TEF (DTRTRS) is called.

*Constraint*: $\mathrm{LDB} \geq \max(1, \mathrm{N})$.

10: INFO – INTEGER *Output*

*On exit*: INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO $< 0$

If INFO $= -i$, the $i$th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO $> 0$

If INFO $= i$, $a(i,i)$ is exactly zero; $A$ is singular and the solution has not been computed.

## 7 Accuracy

The solutions of triangular systems of equations are usually computed to high accuracy. See Higham (1989).

For each right-hand side vector $b$, the computed solution $x$ is the exact solution of a perturbed system of equations $(A+E)x = b$, where

$$|E| \leq c(n)\epsilon|A|,$$

$c(n)$ is a modest linear function of $n$, and $\epsilon$ is the ***machine precision***.

If $\hat{x}$ is the true solution, then the computed solution $x$ satisfies a forward error bound of the form

$$\frac{\|x-\hat{x}\|_\infty}{\|x\|_\infty} \leq c(n)\,\mathrm{cond}(A,x)\epsilon, \qquad \text{provided} \qquad c(n)\,\mathrm{cond}(A,x)\epsilon < 1,$$

where $\mathrm{cond}(A,x) = \left\| |A^{-1}||A||x| \right\|_\infty / \|x\|_\infty$.

Note that $\mathrm{cond}(A,x) \leq \mathrm{cond}(A) = \left\| |A^{-1}||A| \right\|_\infty \leq \kappa_\infty(A)$; $\mathrm{cond}(A,x)$ can be much smaller than $\mathrm{cond}(A)$ and it is also possible for $\mathrm{cond}\left(A^{\mathrm{T}}\right)$ to be much larger (or smaller) than $\mathrm{cond}(A)$.

Forward and backward error bounds can be computed by calling F07THF (DTRRFS), and an estimate for $\kappa_\infty(A)$ can be obtained by calling F07TGF (DTRCON) with NORM = 'I'.

## 8 Further Comments

The total number of floating point operations is approximately $n^2 r$.

The complex analogue of this routine is F07TSF (ZTRTRS).

## 9 Example

This example solves the system of equations $AX = B$, where

$$A = \begin{pmatrix} 4.30 & 0.00 & 0.00 & 0.00 \\ -3.96 & -4.87 & 0.00 & 0.00 \\ 0.40 & 0.31 & -8.02 & 0.00 \\ -0.27 & 0.07 & -5.95 & 0.12 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} -12.90 & -21.50 \\ 16.75 & 14.93 \\ -17.55 & 6.33 \\ -11.04 & 8.09 \end{pmatrix}.$$

### 9.1 Program Text

```
    Program f07tefe

!     F07TEF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: dtrtrs, nag_wp, x04caf
!     .. Implicit None Statement ..
```

```
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
      Character (1), Parameter         :: diag = 'N', trans = 'N'
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, lda, ldb, n, nrhs
      Character (1)                    :: uplo
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:), b(:,:)
!     .. Executable Statements ..
      Write (nout,*) 'F07TEF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, nrhs
      lda = n
      ldb = n
      Allocate (a(lda,n),b(ldb,nrhs))

!     Read A and B from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)(a(i,i:n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)(a(i,1:i),i=1,n)
      End If
      Read (nin,*)(b(i,1:nrhs),i=1,n)

!     Compute solution

!     The NAG name equivalent of dtrtrs is f07tef
      Call dtrtrs(uplo,trans,diag,n,nrhs,a,lda,b,ldb,info)

!     Print solution

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04caf('General',' ',n,nrhs,b,ldb,'Solution(s)',ifail)

      Else
        Write (nout,*) 'A is singular'
      End If

    End Program f07tefe
```

## 9.2 Program Data

```
F07TEF Example Program Data
  4  2                         :Values of N and NRHS
  'L'                          :Value of UPLO
  4.30
 -3.96  -4.87
  0.40   0.31  -8.02
 -0.27   0.07  -5.95   0.12   :End of matrix A
-12.90 -21.50
 16.75  14.93
-17.55   6.33
-11.04   8.09                 :End of matrix B
```

## 9.3 Program Results

```
 F07TEF Example Program Results

 Solution(s)
                1           2
 1       -3.0000    -5.0000
 2       -1.0000     1.0000
 3        2.0000    -1.0000
 4        1.0000     6.0000
```